Newsgroup: it.comp.appl.access
Titolo: aggiungere una riga da codice ad una Casella di riepilogo

Qual'è il comando per aggiungere una riga su una casella di riepilogo
(formata anche da + colonne)?
Io ho trovato questa soluzione:
Dim MyFile, MyName, i
MyName = ""
MyFile = Dir("C:\*.TXT")
Do While MyFile <> ""
        MyName = MyName + MyFile + ";"
    MyFile = Dir    ' Legge la voce successiva.
Loop
Me.Elenco.RowSource = MyName
però a me non piace tantissimo.
Preferirei un comando che interagisce con il  Do While...Loop piuttosto che
alla fine.
Ringrazio in anticipo gli eventuali interessamenti e consigli
PS la routine serve per elencare i files di una directory e gestirli con la
LISTBOX.
--
  Ciao
Andrea


Prova cosě:
Dim MyFile As String
MyFile = Dir("C:\*.TXT")
Do While MyFile <> ""
        Me!Elenco.RowSource = Me!Elenco.RowSource & MyFile & ";"
        MyFile = Dir    ' Legge la voce successiva.
Loop
Me!Elenco.RowSource = Left(Me!Elenco.RowSource, Len(Me!Elenco.RowSource)-1)
' Questa istruzione serve ad eliminare l'ultimo punto e virgola
--
Roberto
---------------------------------------------
il Sito Comune di it.comp.appl.access:
http://www.sitocomune.com
---------------------------------------------

 la LISTBOX.
    Questo metodo pero' non ti permette di mostrare i nomi dei file in
ordine alfabetico.
    L'esempio che segue fa piu' o meno la stessa cosa della tua routine
usando pero' una tabella. Per testarlo basta che piazzi su una nuova
maschera una Casella di riepilogo di nome "cdrFiles" con etichetta di nome
"cdrFiles Etichetta" e incolli il codice nel modulo della maschera. Verra'
creata una tabella di nome "TFILES". QUINDI ATTENZIONE SE NEL TUO MDB NE
ESISTE GIA' UNA CON TALE NOME. La procedura fa uso di DAO.
Option Compare Database: Option Explicit
Const mc_sPathName = "C:\*.TXT"
Private m_sDblQ As String
Private Function ListBoxFilesPopulate( _
  ByVal AcListBoxObject As Access.ListBox _
, ByVal PathName As String _
) As Boolean
  On Error GoTo erh
  Const c_sTblFiles = "TFILES"
  Const c_sFldName = "Name"
  Const c_sFldNameCaption = "Nome"
  Const c_sFldLength = "Length"
  Const c_sFldLengthCaption = "Dimensione"
  Const c_sFldDateTime = "DateTime"
  Const c_sFldDateTimeCaption = "Data ultima modifica"
  Dim sql   As String
  Dim rst   As DAO.Recordset
  Dim sFile As String
  Dim sPath As String
  On Error Resume Next
  sql = "DELETE * FROM [" & c_sTblFiles & "]"
  CurrentDb.Execute sql, DAO.dbFailOnError
  sql _
    = "CREATE TABLE [" & c_sTblFiles & "]" _
    & "(" _
    & " [" & c_sFldName & "] TEXT ( 255 )" _
    & ",[" & c_sFldLength & "] INTEGER" _
    & ",[" & c_sFldDateTime & "] DATE" _
    & ")"
  CurrentDb.Execute sql, DAO.dbFailOnError
  sql _
    = "ALTER TABLE [" & c_sTblFiles & "]" _
    & " ADD COLUMN [" & c_sFldName & "] TEXT ( 255 )"
  CurrentDb.Execute sql, DAO.dbFailOnError
  sql _
    = "ALTER TABLE [" & c_sTblFiles & "]" _
    & " ADD COLUMN [" & c_sFldLength & "] INTEGER"
  CurrentDb.Execute sql, DAO.dbFailOnError
  sql _
    = "ALTER TABLE [" & c_sTblFiles & "]" _
    & " ADD COLUMN [" & c_sFldDateTime & "] DATE"
  CurrentDb.Execute sql, DAO.dbFailOnError
  On Error GoTo erh
  sql _
    = "SELECT" _
    & " [" & c_sTblFiles & "].*" _
    & " FROM [" & c_sTblFiles & "]"
  Set rst _
    = CurrentDb.OpenRecordset( _
        sql, DAO.dbOpenDynaset, DAO.dbFailOnError)
  sPath = Left$(PathName, InStrRev(PathName, "\"))
  sFile = Dir(PathName, vbArchive Or vbNormal)
  Do While Len(sFile) > 0
    With rst
      .AddNew
      .Fields(c_sFldName).Value _
        = sFile
      .Fields(c_sFldLength).Value _
        = FileLen(sPath & sFile)
      .Fields(c_sFldDateTime).Value _
        = FileDateTime(sPath & sFile)
      .Update
    End With
    sFile = Dir
  Loop
  sql _
    = "SELECT" _
    & " [" & c_sTblFiles & "].[" & c_sFldName & "]" _
    & " AS [" & c_sFldNameCaption & "]" _
    & ",Format$(" _
      & "[" & c_sTblFiles & "].[" & c_sFldLength & "]/1024" _
      & ",'#,##0.00 \K\B')" _
    & " AS [" & c_sFldLengthCaption & "]" _
    & ",Format$(" _
      & "[" & c_sTblFiles & "].[" & c_sFldDateTime & "]" _
      & ",'dd/mm/yyyy hh.nn.ss')" _
    & " AS [" & c_sFldDateTimeCaption & "]" _
    & " FROM [" & c_sTblFiles & "]" _
    & " ORDER BY" _
    & " [" & c_sTblFiles & "].[" & c_sFldName & "]"
  With AcListBoxObject
    .Parent.Controls(.Name & " Etichetta").Caption = PathName
    .RowSourceType = "Table/query"
    .RowSource = sql
    .ColumnCount = rst.Fields.Count
    .ColumnHeads = True
  End With
  ListBoxFilesPopulate = True
ext:
  On Error Resume Next
  rst.Close
  Set rst = Nothing
  Exit Function
erh:
  With Err
    MsgBox .Description, vbCritical, "ERR#" & CStr(.Number)
  End With
  ListBoxFilesPopulate = False
  Resume ext
End Function
Private Sub Form_Load()
  If ( _
    ListBoxFilesPopulate(Me.cdrFiles, mc_sPathName) _
    = _
    True _
  ) Then
    ' Ok.
  Else
    ' Ko.
  End If
End Sub
Private Sub Form_Open( _
  ByRef Cancel As Integer)
  m_sDblQ = Chr$(34)
End Sub
--
Ciao :-)
Maurizio
--------
?SPQR(C)
 X
--------